Project: WiFi Temp & Humidity Monitor

In this project, we will configure the **ESP32** as a Soft Access Point (SoftAP). This means it will act like a mini WiFi router that you can connect to with your phone to view a live web dashboard of the **DHT11** readings.

1. Connections

We will use GPIO 4 for the data signal. If you have the 4-pin bare sensor, remember to put a 10k resistor between VCC and Data. If you have the 3-pin module, just connect directly.

ESP32 Pin	DHT11 Pin	Description
3.3V	**VCC**	Power
GND	**GND**	Ground
GPIO 4	**DATA**	Data Signal

2. Libraries that are needed to be downloaded

You need to install these two libraries from the Arduino Library Manager.
Go to Tools > Manage Libraries and install:

1. **DHT sensor library** by Adafruit.
2. **Adafruit Unified Sensor** by Adafruit.

3. Code

How it works

The code uses `WiFi.softAP()` to create a network named "ESP32-DHT-Server". It then starts a WebServer on port 80. When your phone's browser requests the root page (`/`), the ESP32 reads the DHT11 and serves an HTML page with the values.

Steps to connect

1. Connect the DHT11 to the pins listed above.
2. Upload the code below.
3. On your phone, go to WiFi Settings and connect to **ESP32-DHT-Server** (Password: **123456789**).
4. Open a web browser (Chrome/Safari/Edge) and type **192.168.4.1** in the address bar.

Code

```
#include <WiFi.h>
#include <WebServer.h>
#include "DHT.h"

#define DHTPIN 4
#define DHTTYPE DHT11

// Aim: Broadcast WiFi and serve sensor data to mobile
const char* ssid = "ESP32-DHT-Server";
const char* password = "123456789";

DHT dht(DHTPIN, DHTTYPE);
WebServer server(80);

void handleRoot() {
  float h = dht.readHumidity();
  float t = dht.readTemperature();

  // Simple HTML Page
  String html = "<html><head><meta name='viewport' content='width=device-width, initial-
scale=1.0'>";
  html += "<meta http-equiv='refresh' content='5'></head><body style='font-family:Arial;
text-align:center;'>";
  html += "<h1>ESP32 Weather Station</h1>";

  if (isnan(h) || isnan(t)) {
    html += "<p style='color:red;'>Error reading sensor!</p>";
  } else {
    html += "<h2>Temperature: " + String(t) + " &deg;C</h2>";
    html += "<h2>Humidity: " + String(h) + " %</h2>";
  }

  html += "<p>Page refreshes every 5 seconds.</p></body></html>";
  server.send(200, "text/html", html);
}

void setup() {
  Serial.begin(115200);
  dht.begin();

  // Create the Access Point
  WiFi.softAP(ssid, password);

  Serial.println("Access Point Started");
  Serial.print("IP Address: ");
  Serial.println(WiFi.softAPIP());

  // Define what happens when someone visits the IP
  server.on("/", handleRoot);
  server.begin();
}
```

```
void loop() {
  server.handleClient(); // Keep the web server running
}
```

4. What to expect

- ✓ After uploading, your ESP32 becomes a WiFi Hotspot.
- ✓ Once connected via mobile, the page at **192.168.4.1** will show the current temperature and humidity.
- ✓ The page includes a "meta refresh" tag, so it will automatically update every 5 seconds without you needing to hit reload.